Résolution d'une équation cubique complexe

Le but de cette page de calcul est de montrer comment on peut résoudre, grâce à Tartaglia, une équation cubique complexe de la forme

```
var('u','v','y','z','p','q') # déclaration des variables dont on aura besoin
eqc  = z^3+p*z+q == 0
eqc
```

$$z^3 + pz + q = 0$$

où p et q sont des nombres complexes donnés. Remarquons que le symbole `=` dans sage est l'opérateur de définition et que le symbole `==` désigne la relation d'égalité. Ici on manipule donc des équations: la valeur de la variable `eqc` est l'équation `z^3 + p * z + q == 0`, et non pas l'expression formelle `z^3 + p * z + q`

Pour résoudre l'équation en question, on suppose que z en est une solution. La méthode de Tartaglia consiste à faire la substitution $z = u + v$, où u et v sont des nombres complexes à déterminer par la suite:

```
eqc.subs(z=u+v)
```

$$(u + v)^3 + p(u + v) + q = 0$$

```
_.expand()
```

$$u^3 + 3\,u^2v + 3\,uv^2 + v^3 + pu + pv + q = 0$$

L'idée de génie de Tartaglia était de supposer que $3uv + p = 0$. Cela simplifiera considérablement l'équation ci-dessus, surtout lorsque $p \neq 0$. En effet, supposons que $3uv + p = 0$ et $p \neq 0$. Dans ce cas $u \neq 0$ et on peut écrire $v = -p/(3u)$. Substituons $v = -p/(3u)$ dans l'équation ci-dessus:

```
_.subs(v=-p/(3*u))
```

$$u^3 + q - \frac{p^3}{27\,u^3} = 0$$

Multiplions par u^3:

```
(_*u^3).expand()
```

$$u^6 + qu^3 - \frac{1}{27}\,p^3 = 0$$

C'est une expression quadratique en u^3! Pour résumer, si $z^3 + pz + q = 0$ et si on pose $z = u + v$ et $v = -p/(3u)$, on a que u^3 est solution de l'équation quadratique suivante en y.

```
eqq=_.subs(u=y^(1/3))
```

```
eqq
```

$$-\frac{1}{27}\,p^3 + qy + y^2 = 0$$

qu'on sait résoudre. Soient a, b, c les coefficients de $y^2, y, 1$, respectivement:

```
a=eqq.lhs().coeff(y,2)  # lhs() calcule le premier membre de l'équation eqq
b=eqq.lhs().coeff(y,1)  # coeff(y,n) donne le coefficient de y^n
c=eqq.lhs().coeff(y,0)
a,b,c
```

$$\left(1, q, -\frac{1}{27}\,p^3\right)$$

Le discriminant Δ de l'équation quadratique est égal à:

```
Delta=b^2-4*a*c
Delta
```

$$\frac{4}{27}\,p^3 + q^2$$

Soit δ une racine carrée de Δ:

```
delta=sqrt(Delta)
delta
```

$$\sqrt{\frac{4}{27}\,p^3 + q^2}$$

Voilà une notation qu'il faut bien comprendre: $\sqrt{w}$ signifie «une racine carrée de w». C'est donc une expression formelle qui n'a pas d'autre propriété que $\left(\sqrt{w}\right)^2 = w$. Si elle figure à plusieurs reprises dans une expression, elle est sensée être la même racine carrée de w, sans qu'on sache laquelle c'est.

Soit y_1 l'une des racines de l'équation quadratique ci-dessus. Elle nous mènera à toutes les solutions de l'équation cubique du départ. On pourra vérifier que l'autre racine y_2 de l'équation quadratique donne lieu aux mêmes solutions de l'équation cubique.

```
y1=(-b+delta)/2
y1
```

$$-\frac{1}{2}\,q + \frac{1}{2}\,\sqrt{\frac{4}{27}\,p^3 + q^2}$$

Vérifions que y_1 est effectivement racine de l'équation quadratique:

```
eqq.subs(y=y1)
```

$$-\frac{1}{27}\,p^3+\frac{1}{4}\left(q-\sqrt{\frac{4}{27}\,p^3+q^2}\right)^2-\frac{1}{2}\left(q-\sqrt{\frac{4}{27}\,p^3+q^2}\right)q=0$$

```
bool(_)
```

True

Remarquons que le fonction `bool()`évalue la valeur booléenne de son argument, ici la relation `y1^3+p*y1+q==0`

Soient u_1, u_2, u_3 les trois racines cubiques de y_1 dans $\mathbf{C}$:

```
u1=y1^(1/3); u2=u1*(-1/2+1/2*I*sqrt(3)); u3=u1*(-1/2-
1/2*I*sqrt(3))
```

On pose encore $v_i=-p/(3u_i)$ et $z_i=u_i+v_i$, pour $i=1,2,3$.

```
v1=-p/3/u1; v2=-p/3/u2; v3=-p/3/u3
z1=u1+v1; z2=u2+v2; z3=u3+v3
```

Les nombres complexes z_1, z_2, z_3 sont les solutions de l'équation cubique initiale:

```
bool(eqc.subs(z=z1))
```

True

```
bool(eqc.subs(z=z2))
```

True

```
bool(eqc.subs(z=z3))
```

True

En fait, ce sont toutes les solutions de l'équation $z^3+pz+q=0$ car

```
bool((z-z1)*(z-z2)*(z-z3) == eqc.lhs())
```

True

Explicitement, les solutions sont:

```
z1
```

$$-\frac{p}{3\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}}+\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}$$

```
z2
```

$$-\frac{1}{2}\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}\left(-i\sqrt{3}+1\right)+\frac{2p}{3\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}(-}$$

```
z3
```

$$-\frac{1}{2}\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}\left(i\sqrt{3}+1\right)+\frac{2p}{3\left(-\frac{1}{2}q+\frac{1}{2}\sqrt{\frac{4}{27}p^3+q^2}\right)^{\frac{1}{3}}(i\sqrt{3}}$$

On peut maintenant calculer des exemples explicites:

Les solutions de l'équation $z^3+z+1=0$ sont les nombres complexes:

```
z1.subs(p=1,q=1).simplify_radical()
```

$$-\frac{\sqrt{3}2^{\frac{2}{3}}\left(2^{\frac{2}{3}}-\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{2}{3}}\right)}{6\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{1}{3}}}$$

```
z2.subs(p=1,q=1).simplify_radical()
```

$$\frac{\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{2}{3}}\left(i\sqrt{3}+1\right)+2\,2^{\frac{2}{3}}}{\left(\sqrt{3}2^{\frac{1}{3}}-3i\,2^{\frac{1}{3}}\right)\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{1}{3}}}$$

```
z3.subs(p=1,q=1).simplify_radical()
```

$$-\frac{\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{2}{3}}\left(i\sqrt{3}-1\right)-2\,2^{\frac{2}{3}}}{\left(\sqrt{3}2^{\frac{1}{3}}+3i\,2^{\frac{1}{3}}\right)\left(\sqrt{31}-3\sqrt{3}\right)^{\frac{1}{3}}}$$

Les solutions de l'équation $z^3-3z+1=0$ sont les nombres complexes:

```
z1.subs(p=-3,q=1).simplify_radical()
```

$$\frac{2^{\frac{2}{3}}\left(2^{\frac{2}{3}}+\left(i\sqrt{3}-1\right)^{\frac{2}{3}}\right)}{2\left(i\sqrt{3}-1\right)^{\frac{1}{3}}}$$

```
z2.subs(p=-3,q=1).simplify_radical()
```

$$-\frac{\left(i\sqrt{3}+1\right)\left(i\sqrt{3}-1\right)^{\frac{2}{3}}-2\,2^{\frac{2}{3}}}{\left(i\sqrt{3}2^{\frac{1}{3}}-2^{\frac{1}{3}}\right)\left(i\sqrt{3}-1\right)^{\frac{1}{3}}}$$

```
z3.subs(p=-3,q=1).simplify_radical()
```

$$-\frac{\left(i\sqrt{3}-1\right)^{\frac{5}{3}}+2\,2^{\frac{2}{3}}}{\left(i\sqrt{3}2^{\frac{1}{3}}+2^{\frac{1}{3}}\right)\left(i\sqrt{3}-1\right)^{\frac{1}{3}}}$$

Remarquons que, malgré les apparances, ces 3 nombres sont tous réels! Ici, ils sont donnés sous format radicale complexe. En fait, on peut démontrer qu'on ne peut les représenter sous forme radical réelle, i.e., faisant intervenir que les symboles $+, -, x, /, \cdot^n, \sqrt[n]{\cdot}$ et des nombres réels, où n peut être n'importe quel nombre naturel. Par contre, il existe une représentation trigonométrique réelle des solutions de l'équation $z^3 - 3z + 1 = 0$. On renvoit à la feuille de calcul tartaglia3.